Computation of the Control law using pole placement to stabilize an inverted pendulum on a moving cart

by Nasser Abbasi, computation section project report for 511

set up some notations to use

```
<< Notation`
Symbolize[ x₁ ]; Symbolize[ x₂ ]; Symbolize[ x₃ ];
Symbolize[ x₄ ]; Symbolize[ vₓ ]; Symbolize[ v_y ]
```

Calculate the kinetic energy

```
ke = 1/2 M x'[t]^2 + 1/2 m vSquared;
vSquared = (x'[t] + vₓ[t])^2 + v_y[t]^2;
vSquared = vSquared /. vₓ[t] → v[t] Cos[θ[t]];
vSquared = vSquared /. v_y[t] → v[t] Sin[θ[t]];
vSquared = vSquared /. v[t] → L θ'[t];
vSquared = Simplify[vSquared];
vSquared = Expand[vSquared];
ke
```

$$\frac{1}{2} m \left(L^2 \theta'(t)^2 + 2 L x'(t) \theta'(t) \cos(\theta(t)) + x'(t)^2\right) + \frac{1}{2} M x'(t)^2$$

Calculate the potential energy of the system

```
pe = m g L Cos[θ[t]]
```

$$g L m \cos(\theta(t))$$

Calculate the Lagrangian of system

```
Lagrangian = ke - pe
```

$$-g\,L\,m\cos(\theta(t))+\frac{1}{2}\,m\left(L^2\,\theta'(t)^2+2\,L\,x'(t)\,\theta'(t)\cos(\theta(t))+x'(t)^2\right)+\frac{1}{2}\,M\,x'(t)^2$$

Find equation of motion for the bob

```
eqForTheta = D[D[Lagrangian, θ'[t]], t] - D[Lagrangian, θ[t]]
eqForTheta = Assuming[{L, m} > 0, Simplify[eqForTheta]];
eqForTheta = Cancel[eqForTheta * (1 / (L m))] == 0
```

$$-g\,L\,m\sin(\theta(t))+\frac{1}{2}\,m\left(2\,L^2\,\theta''(t)+2\,L\,x''(t)\cos(\theta(t))-2\,L\,x'(t)\,\theta'(t)\sin(\theta(t))\right)+L\,m\,x'(t)\,\theta'(t)\sin(\theta(t))$$

$$-g\sin(\theta(t))+L\,\theta''(t)+x''(t)\cos(\theta(t))=0$$

Find equation of motion for the bob for small angle

```
eqForThetaForSmallAngle = eqForTheta /. {Cos[θ[t]] → 1, Sin[θ[t]] -> θ[t], θ'[t]^2 → 0}
```

$$-g\,\theta(t)+L\,\theta''(t)+x''(t)=0$$

Find equation of motion for the cart

```
eqForX = D[D[Lagrangian, x'[t]], t] - D[Lagrangian, x[t]]
eqForX = Simplify[eqForX == u ]
```

$$\frac{1}{2}\,m\left(2\,L\,\theta''(t)\cos(\theta(t))-2\,L\,\theta'(t)^2\sin(\theta(t))+2\,x''(t)\right)+M\,x''(t)$$

$$L\,m\,\theta'(t)^2\sin(\theta(t))+u=L\,m\,\theta''(t)\cos(\theta(t))+(m+M)\,x''(t)$$

Find equation of motion for the cart for small angle

```
eqForXForSmallAngle = eqForX /. {Cos[θ[t]] → 1, Sin[θ[t]] -> θ[t], θ'[t]^2 → 0}
```

$$u = L\,m\,\theta''(t) + (m + M)\,x''(t)$$

```
First[Solve[eqForThetaForSmallAngle, θ''[t]]]
```

$$\left\{\theta''(t) \to \frac{g\,\theta(t) - x''(t)}{L}\right\}$$

```
First[Solve[eqForXForSmallAngle, x''[t]]]
```

$$\left\{x''(t) \to \frac{u - L\,m\,\theta''(t)}{m + M}\right\}$$

```
sol = First[Solve[{eqForXForSmallAngle, eqForThetaForSmallAngle}, {x''[t], θ''[t]}]]
```

$$\left\{x''(t) \to -\frac{g\,m\,\theta(t) - u}{M},\ \theta''(t) \to -\frac{-g\,m\,\theta(t) - g\,M\,\theta(t) + u}{L\,M}\right\}$$

Set up the state space equations X' = A X + B u

```
x1dot = x2 == 0;
x2dot = ((x''[t] /. sol) /. {θ[t] → x3}) == 0;
x3dot = x4 == 0;
x4dot = ((θ''[t] /. sol) /. {θ[t] → x3}) == 0;
c = CoefficientArrays[{x1dot, x2dot, x3dot, x4dot}, {x1, x2, x3, x4}] // Normal;
```

setup the A matrix

```
A = c[[2]]
```

$$\begin{pmatrix} 0 & 1 & 0 & 0 \\ 0 & 0 & -\frac{g\,m}{M} & 0 \\ 0 & 0 & 0 & 1 \\ 0 & 0 & \frac{g}{L} + \frac{m\,g}{L\,M} & 0 \end{pmatrix}$$

setup the B matrix

```
b = Transpose[{c[[1]] /. u → 1}]
```

$$\begin{pmatrix} 0 \\ \frac{1}{M} \\ 0 \\ -\frac{1}{L M} \end{pmatrix}$$

Analysis for Uncontrolled inverted pendulum

In this section, we find the solutions (x (t), x' (t), θ (t), and θ' (t)) when no control is applied. i.e. u = 0

define initial conditions, and initial X(0) vector, and define system values

```
values = {initialXPosition → 0, initialXSpeed → 1,
   initialAngle → Pi / 20, initialAngularSpeed → 1, m → 0.1, M → 1, g → 10, L → 1};
MatrixForm[initialX = {initialXPosition, initialXSpeed,
    initialAngle, initialAngularSpeed} /. values]
```

$$\begin{pmatrix} 0 \\ 1 \\ \frac{\pi}{20} \\ 1 \end{pmatrix}$$

define desired pole locations

```
desiredPoleLocations = {-1, -2, -1 + I, -1 - I}
```

$\{-1, -2, -1 + i, -1 - i\}$

Setup the sI - A matrix

```
tmp = s * IdentityMatrix[4] - A
```

$$\begin{pmatrix} s & -1 & 0 & 0 \\ 0 & s & \frac{g m}{M} & 0 \\ 0 & 0 & s & -1 \\ 0 & 0 & -\frac{g}{L} - \frac{m g}{L M} & s \end{pmatrix}$$

Find Inverse Laplace transform of (sI – A)$^{-1}$

```
Inverse[tmp /. values];
InverseLaplaceTransform[%, s, t]
```

$$\begin{pmatrix} 1. & 1.\,t & -1.\,(-0.0909091 + 0.0454545\,e^{-3.31662\,t} + 0.0454545\,e^{3.31662\,t}) & -1.\,(-0.0909091\,t - 0.0137051\,e^{-3.31662\,t} + 0.0137051\,e^{3 \ldots} \\ 0 & 1. & -1.\,(-0.150756\,e^{-3.31662\,t} + 0.150756\,e^{3.31662\,t}) & -1.\,(-0.0909091 + 0.0454545\,e^{-3.31662\,t} + 0.0454545\,e^{3. \ldots} \\ 0 & 0 & 0.5\,e^{-3.31662\,t} + 0.5\,e^{3.31662\,t} & -0.150756\,e^{-3.31662\,t} + 0.150756\,e^{3.31662\,t} \\ 0 & 0 & 11.\,(-0.150756\,e^{-3.31662\,t} + 0.150756\,e^{3.31662\,t}) & 0.5\,e^{-3.31662\,t} + 0.5\,e^{3.31662\,t} \end{pmatrix}$$

Find X (t) the solution which is InverseLaplaceTransform of (sI – A)$^{-1}$ * X (0)

```
TableForm[solForUncontrolled = %.initialX]
```

$$-0.15708\,(0.0454545\,e^{-3.31662\,t} + 0.0454545\,e^{3.31662\,t} - 0.0909091) - 1.\,(-0.0909091\,t - 0.0137051\,e^{-3.31662\,t} + 0.0137051\,e^{3.31662\,t} \ldots$$

$$-1.\,(0.0454545\,e^{-3.31662\,t} + 0.0454545\,e^{3.31662\,t} - 0.0909091) - 0.15708\,(0.150756\,e^{3.31662\,t} - 0.150756\,e^{-3.31662\,t}) + 1.$$

$$\frac{1}{20}\,\pi\,(0.5\,e^{-3.31662\,t} + 0.5\,e^{3.31662\,t}) - 0.150756\,e^{-3.31662\,t} + 0.150756\,e^{3.31662\,t}$$

$$1.72788\,(0.150756\,e^{3.31662\,t} - 0.150756\,e^{-3.31662\,t}) + 0.5\,e^{-3.31662\,t} + 0.5\,e^{3.31662\,t}$$

Plot the solutions $x_1(t)$, $x_2(t)$, $x_3(t)$, $x_4(t)$

```
FullSimplify[solForUncontrolled[[1]]];
xSolutionForUncontolled = ExpToTrig[%];
p1 = Plot[% , {t, 0, 1}, AxesLabel → {t, "x(t) degrees"},
   PlotLabel → "x(t), no control", ImageSize → 240];

FullSimplify[solForUncontrolled[[2]]];
xDotSolutionForUncontolled = ExpToTrig[%];
p2 = Plot[% , {t, 0, 1}, AxesLabel → {t, "x'(t) degrees"},
   PlotLabel → "x'(t), no control", ImageSize → 240];

FullSimplify[solForUncontrolled[[3]]];
thetaSolutionForUncontolled = ExpToTrig[%];
p3 = Plot[180 / Pi * % , {t, 0, 1}, AxesLabel → {t, "θ(t) degrees"},
   PlotLabel → "θ(t), no control", ImageSize → 240];

FullSimplify[solForUncontrolled[[4]]];
thetaDotSolutionForUncontolled = ExpToTrig[%];
p4 = Plot[180 / Pi * % , {t, 0, 1}, AxesLabel → {t, "θ'(t) degrees/sec"},
   PlotLabel → "θ'(t), no control", ImageSize → 240];

GraphicsGrid[{{p1, p2}, {p3, p4}}]
```



Analysis for Controlled inverted pendulum

In this section, we find the solutions (x (t), x' (t), θ (t), and θ' (t)) when control force is applied which is first determined based on desired pole locations

Setup the A matrix with the control law. See analytical report for how this is derived

```
AForControlled =
 {{0, 1, 0, 0}, {f1/M, f2/M, f3/M - g m/M, f4/M}, {0, 0, 0, 1}, {-f1/(M L), -f2/(M L), g (m + M)/(M L) - f3/(M L), -f4/(M L)}}
```

$$\begin{pmatrix} 0 & 1 & 0 & 0 \\ \frac{f1}{M} & \frac{f2}{M} & \frac{f3}{M} - \frac{g\,m}{M} & \frac{f4}{M} \\ 0 & 0 & 0 & 1 \\ -\frac{f1}{L\,M} & -\frac{f2}{L\,M} & \frac{g\,(m+M)}{L\,M} - \frac{f3}{L\,M} & -\frac{f4}{L\,M} \end{pmatrix}$$

Find the characteristic equation for the A matrix

```
tmp = s * IdentityMatrix[4] - AForControlled;
givenCharEquation = Det[tmp]
```

$$\frac{f1\,g}{L\,M} - \frac{f1\,s^2}{M} + \frac{f2\,g\,s}{L\,M} - \frac{f2\,s^3}{M} + \frac{f3\,s^2}{L\,M} + \frac{f4\,s^3}{L\,M} - \frac{g\,m\,s^2}{L\,M} - \frac{g\,s^2}{L} + s^4$$

Determine the desired characteristic equation from the desired closed loop pole locations

```
desiredCharEquation =
 Expand[Apply[Times, Table[s - desiredPoleLocations[[i]], {i, 1, 4}]]]
```

$$s^4 + 5\,s^3 + 10\,s^2 + 10\,s + 4$$

Compare coefficients of the above 2 characteristic equations to solve for f1,f2,f3,f4

```
tmp = Collect[desiredCharEquation - givenCharEquation, s^_]
```

$$s^2\left(\frac{f1}{M} - \frac{f3}{L\,M} + \frac{g\,m}{L\,M} + \frac{g}{L} + 10\right) - \frac{f1\,g}{L\,M} + s^3\left(\frac{f2}{M} - \frac{f4}{L\,M} + 5\right) + s\left(10 - \frac{f2\,g}{L\,M}\right) + 4$$

```
tmp = CoefficientList[tmp, s]
```

$$\left\{4-\frac{f1\,g}{L\,M},\ 10-\frac{f2\,g}{L\,M},\ \frac{f1}{M}-\frac{f3}{L\,M}+\frac{g\,m}{L\,M}+\frac{g}{L}+10,\ \frac{f2}{M}-\frac{f4}{L\,M}+5\right\}$$

```
equations = Table[tmp[[i]] == 0, {i, 1, 4}]
```

$$\left\{4-\frac{f1\,g}{L\,M}=0,\ 10-\frac{f2\,g}{L\,M}=0,\ \frac{f1}{M}-\frac{f3}{L\,M}+\frac{g\,m}{L\,M}+\frac{g}{L}+10=0,\ \frac{f2}{M}-\frac{f4}{L\,M}+5=0\right\}$$

```
theF = First[Solve[equations, {f1, f2, f3, f4}]]
```

$$\left\{f3\to-\frac{g^2\,(-m)-g^2\,M-10\,g\,L\,M-4\,L^2\,M}{g},\ f4\to\frac{5\,(g\,L\,M+2\,L^2\,M)}{g},\ f1\to\frac{4\,L\,M}{g},\ f2\to\frac{10\,L\,M}{g}\right\}$$

Now that we have solved for the f' s, we plug in the numerical values to obtain numerical value for the f' s

```
TableForm[theF /. values // N]
```

$f3\to 21.4$
$f4\to 6.$
$f1\to 0.4$
$f2\to 1.$

Now we can update the A matrix with the values found for the control law

```
AForControlled = AForControlled /. theF
```

$$\begin{pmatrix} 0 & 1 & 0 & 0 \\ \frac{4\,L}{g} & \frac{10\,L}{g} & -\frac{g\,m}{M}-\frac{-m\,g^2-M\,g^2-10\,L\,M\,g-4\,L^2\,M}{g\,M} & \frac{5\left(2\,M\,L^2+g\,M\,L\right)}{g\,M} \\ 0 & 0 & 0 & 1 \\ -\frac{4}{g} & -\frac{10}{g} & \frac{g\,(m+M)}{L\,M}+\frac{-m\,g^2-M\,g^2-10\,L\,M\,g-4\,L^2\,M}{g\,L\,M} & -\frac{5\left(2\,M\,L^2+g\,M\,L\right)}{g\,L\,M} \end{pmatrix}$$

```
AForControlled = AForControlled /. values
```

$$\begin{pmatrix} 0 & 1 & 0 & 0 \\ \frac{2}{5} & 1 & 20.4 & 6 \\ 0 & 0 & 0 & 1 \\ -\frac{2}{5} & -1 & -10.4 & -6 \end{pmatrix}$$

Now we can solve the system by method of inverse Laplace transform

```
tmp = s * IdentityMatrix[4] - AForControlled
```

$$\begin{pmatrix} s & -1 & 0 & 0 \\ -\frac{2}{5} & s-1 & -20.4 & -6 \\ 0 & 0 & s & -1 \\ \frac{2}{5} & 1 & 10.4 & s+6 \end{pmatrix}$$

```
Inverse[tmp]
```

$$\begin{pmatrix} \frac{s^3+5 s^2+10.4 s+10.}{s^4+5 s^3+10. s^2+10. s+4.} & \frac{s^2+6 s+10.4}{s^4+5 s^3+10. s^2+10. s+4.} & \frac{20.4 s+60.}{s^4+5 s^3+10. s^2+10. s+4.} & \frac{6 s+20.4}{s^4+5 s^3+10. s^2+10. s+4.} \\ \frac{\frac{2 s^2}{5}-4.}{s^4+5 s^3+10. s^2+10. s+4.} & \frac{s^3+6 s^2+10.4 s}{s^4+5 s^3+10. s^2+10. s+4.} & \frac{20.4 s^2+60. s}{s^4+5 s^3+10. s^2+10. s+4.} & \frac{6 s^2+20.4 s}{s^4+5 s^3+10. s^2+10. s+4.} \\ -\frac{2 s}{5\left(s^4+5 s^3+10. s^2+10. s+4.\right)} & \frac{-s-\frac{2}{5}}{s^4+5 s^3+10. s^2+10. s+4.} & \frac{s^3+5 s^2-\frac{2 s}{5}}{s^4+5 s^3+10. s^2+10. s+4.} & \frac{s^2-s-\frac{2}{5}}{s^4+5 s^3+10. s^2+10. s+4.} \\ -\frac{2 s^2}{5\left(s^4+5 s^3+10. s^2+10. s+4.\right)} & \frac{-s^2-\frac{2 s}{5}}{s^4+5 s^3+10. s^2+10. s+4.} & \frac{-10.4 s^2-10. s-4.}{s^4+5 s^3+10. s^2+10. s+4.} & \frac{s^3-s^2-\frac{2 s}{5}}{s^4+5 s^3+10. s^2+10. s+4.} \end{pmatrix}$$

```
InverseLaplaceTransform[%, s, t];
TableForm[solForControlled = %.initialX /. values]
```

$e^{(-1.-1. i) t}((-5.1-2.1 i)-(5.1-2.1 i) e^{2. i t})+e^{(-1.-1. i) t}((-2.1-0.1 i)-(2.1-0.1 i) e^{2. i t})+\frac{1}{20} \pi (e^{(-1.-1. i) t}((-15.-4.8 i)-(15. -$

$e^{(-1.-1. i) t}((2.-2.2 i) e^{2. i t}+(2.+2.2 i))+e^{(-1.-1. i) t}((3.-7.2 i) e^{2. i t}+(3.+7.2 i))+\frac{1}{20} \pi (e^{(-1.-1. i) t}((10.2-19.8 i) e^{2. i t}+(10.2 +$

$e^{(-1.-1. i) t}((0.1+0.4 i) e^{2. i t}+(0.1-0.4 i))+e^{(-1.-1. i) t}((0.6+0.9 i) e^{2. i t}+(0.6-0.9 i))+\frac{1}{20} \pi (e^{(-1.-1. i) t}((1.5+2.7 i) e^{2. i t}+(1.5 -$

$e^{(-1.-1. i) t}((-1.5+0.3 i)-(1.5+0.3 i) e^{2. i t})+e^{(-1.-1. i) t}((-0.5+0.3 i)-(0.5+0.3 i) e^{2. i t})+\frac{1}{20} \pi (e^{(-1.-1. i) t}((-4.2+1.2 i)-(4.2 +$

Now that we have found the solutions, we plot them and compare the result for the uncontrolled case

```
FullSimplify[solForControlled[[1]]];
xSolutionForContolled = ExpToTrig[%];
p1 = Plot[% , {t, 0, 10}, AxesLabel → {t, "x(t) degrees"},
   PlotLabel → "x(t) controlled", ImageSize → 240, PlotRange → All];

FullSimplify[solForControlled[[2]]];
xDotSolutionForContolled = ExpToTrig[%];
p2 = Plot[% , {t, 0, 10}, AxesLabel → {t, "x'(t) degrees"},
   PlotLabel → "x'(t) controlled", ImageSize → 240, PlotRange → All];

FullSimplify[solForControlled[[3]]];
thetaSolutionForContolled = ExpToTrig[%];
p3 = Plot[180 / Pi * % , {t, 0, 10}, AxesLabel → {t, "θ(t) degrees"},
   PlotLabel → "θ(t) controlled", ImageSize → 240, PlotRange → All];

FullSimplify[solForControlled[[4]]];
thetaDotSolutionForContolled = ExpToTrig[%];
p4 = Plot[180 / Pi * % , {t, 0, 10}, AxesLabel → {t, "θ'(t) degrees/sec"},
   PlotLabel → "θ'(t)controlled", ImageSize → 240, PlotRange → All];

GraphicsGrid[{{p1, p2}, {p3, p4}}]
```